Externe Anwendungen starten und kontrollieren, Teil 1
Versionen: Access 2003, 2002/XP, 2000 und 97

In Ihren VBA-Lösungen müssen Sie immer wieder externe Anwendungen starten, um beispielsweise den FTP-Transfer von Daten von einem Großrechner durchzuführen, die anschließend in Access weiterverarbeitet werden sollen. Häufig gestaltet sich dieses Zusammenspiel recht kompliziert: Hat die externe Anwendung ihre Arbeit schon erledigt? Stehen die Daten zur weiteren Verarbeitung komplett zur Verfügung? Die Shell-Funktion ist hier wenig hilfreich, da sie lediglich externe Anwendungen starten kann, aber keine Statusabfragen erlaubt. Und lassen sich die oben gestellten Fragen nicht eindeutig beantworten, kann es zu ernsthaften Problemen kommen: Zugriffskonflikte beim Öffnen der Datendatei, Verarbeitung alter und somit falscher Inhalte oder nicht abgefangene Laufzeit-Fehler, um nur einige zu nennen.

Mit diesem zweiteiligen Tipp der Woche stellen wir Ihnen eine Klassenbibliothek vor, die genau diese Probleme löst: Sie erlaubt den Start einer externen Anwendung und die Abfrage, ob diese Anwendung noch ausgeführt wird. Im ersten Teil legen wir die Klassenbibliothek an, der zweite Teil erläutert den Einsatz der Klassenbibliothek in der Praxis und die Funktionsweise.

Um die Klassenbibliothek anzulegen und die Deklarationen, Funktionen und Prozeduren einzugeben, gehen Sie wie folgt vor:

1. Öffnen Sie die Datenbank mit der Lösung, aus der heraus eine externe Anwendung gestartet und kontrolliert werden soll.

2. Wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung (nur Access 2003, 2002/XP, 2000).

3. Wählen Sie das Menü Einfügen-Klassenmodul an.

4. Drücken Sie F4, um das Eigenschaftenfenster anzeigen zu lassen und benennen Sie das Modul "clsExecProg" (Access 2003, 2002/XP, 2000) bzw. speichern Sie das Sie das Modul unter dem Namen "clsExecProg" (Access 97)

5. Geben Sie im allgemeinen Teil des Moduls die folgenden Deklarationen ein bzw. kopieren Sie sie aus dem Newsletter in den allgemeinen Teil:

Option Explicit

'Typen-Deklarationen...
Private Type STARTUPINFO
  CB As Long
  lpReserved As String
  lpDesktop As String
  lpTitle As String
  dwX As Long
  dwY As Long
  dwXSize As Long
  dwYSize As Long
  dwXCountChars As Long
  dwYCountChars As Long
  dwFillAttribute As Long
  dwFlags As Long
  wShowWindow As Integer
  cbReserved2 As Integer
  lpReserved2 As Long
  hStdInput As Long
  hStdOutput As Long
  hStdError As Long
End Type

Private Type PROCESS_INFORMATION
   hProcess As Long
   hThread As Long
   dwProcessID As Long
   dwThreadID As Long
End Type

Private Type SHELLEXECUTEINFO
  cbSize As Long
  fMask As Long
  hwnd As Long
  lpVerb As String
  lpFile As String
  lpParameters As String
  lpDirectory As String
  nShow As Long
  hInstApp As Long
  ' Optionale Felder
  lpIDList As Long
  lpClass As String
  hkeyClass As Long
  dwHotKey As Long
  hIcon As Long
  hProcess As Long
End Type


'API -Funktionen...
Private Declare Function WaitForSingleObject _
                Lib "kernel32" _
               (ByVal hHandle As Long, _
                ByVal dwMilliseconds As Long) As Long

Private Declare Function ShellExecuteEx Lib _
                "shell32.dll" _
                Alias "ShellExecuteExA" ( _
                lpExecInfo As SHELLEXECUTEINFO) As Boolean


'Konstanten...
Private Const NORMAL_PRIORITY_CLASS = &H20&
Private Const INFINITE = -1&
Private Const SEE_MASK_NOCLOSEPROCESS = &H40&
Private Const SEE_MASK_FLAG_NO_UI = &H400&

Private Const STARTF_FORCEOFFFEEDBACK = &H80
Private Const STARTF_FORCEONFEEDBACK = &H40
Private Const STARTF_RUNFULLSCREEN = &H20
Private Const STARTF_USECOUNTCHARS = &H8
Private Const STARTF_USEFILLATTRIBUTE = &H10
Private Const STARTF_USEPOSITION = &H4
Private Const STARTF_USESHOWWINDOW = &H1
Private Const STARTF_USESIZE = &H2
Private Const STARTF_USESTDHANDLES = &H100

'Interne Variablen...
Private ep_ProcInfo As PROCESS_INFORMATION
Private ep_StartupInfo As STARTUPINFO
Private ep_ShellExecInfo As SHELLEXECUTEINFO

'Eigenschaften...
'Programm -> X:\Pfad\Datei.exe|com|bat etc.
Private ep_ProgName As String

'Parameter -> z.B. Dokument
Private ep_ProgParas As String

Private ep_WorkDir As String 'Arbeitsverzeichnis
Private ep_Error As Boolean 'Fehler aufgetreten?

'Fenster-Modus für Programm
Private ep_WinMode As Long
· Geben Sie danach die folgenden Prozeduren und Funktionen ein bzw. kopieren Sie sie aus dem Newsletter in das Modul:

Public Property Let WinMode(lngWinMode As Long)

  ep_WinMode = lngWinMode

End Property

Private Sub Class_Initialize()

  ep_Error = False 'Kein Fehler
  ep_WinMode = vbMaximizedFocus 'Vollbild
  ep_WorkDir = Environ$("windir") 'Windows-Verzeichnis

End Sub

Public Property Let ProgParas(strParas As String)

  ep_ProgParas = strParas

End Property

Public Sub ExecProg()

  If ep_ProgName <> "" Then 'Anwendung angegeben...
    ep_ShellExecInfo.cbSize = Len(ep_ShellExecInfo)
    ep_ShellExecInfo.fMask = SEE_MASK_NOCLOSEPROCESS + _
                             SEE_MASK_FLAG_NO_UI
    ep_ShellExecInfo.lpFile = ep_ProgName
    ep_ShellExecInfo.lpParameters = ep_ProgParas
    ep_ShellExecInfo.lpDirectory = ep_WorkDir
    ep_ShellExecInfo.nShow = ep_WinMode
    'Default: vbNormalFocus
    ep_Error = ShellExecuteEx(ep_ShellExecInfo)
    ep_ProcInfo.hProcess = ep_ShellExecInfo.hProcess
    ep_Error = (ep_ShellExecInfo.hInstApp < 32)
  Else 'Nur Datei/Dokument angegeben...
    ep_ShellExecInfo.cbSize = Len(ep_ShellExecInfo)
    ep_ShellExecInfo.fMask = SEE_MASK_NOCLOSEPROCESS + _
                             SEE_MASK_FLAG_NO_UI
    ep_ShellExecInfo.lpFile = ep_ProgParas
    ep_ShellExecInfo.lpParameters = vbNullString
    ep_ShellExecInfo.lpDirectory = ep_WorkDir
    ep_ShellExecInfo.nShow = ep_WinMode 
    'Default: vbNormalFocus
    ep_Error = ShellExecuteEx(ep_ShellExecInfo)
    ep_ProcInfo.hProcess = ep_ShellExecInfo.hProcess
    ep_Error = (ep_ShellExecInfo.hInstApp < 32)
  End If

End Sub

Public Property Get IsRunning() As Boolean
  Dim lngStatus As Long

  lngStatus = WaitForSingleObject( _
              ep_ProcInfo.hProcess, 10)

  If lngStatus = -1 Then 'Ungültiger Handle...
    IsRunning = False
    ep_Error = True
  Else
    IsRunning = (lngStatus <> 0)
  End If

End Property

Public Property Get EPError() As Boolean

  EPError = ep_Error

End Property

Public Property Let ProgName(strProgName As String)

  ep_ProgName = strProgName

End Property

Public Property Let WorkDir(strWorkDir As String)

  ep_WorkDir = strWorkDir

End Property
· Speichern Sie die Änderungen und verlassen Sie die VBA-Entwicklungsumgebung wieder.

Weiter geht es im zweiten Teil dieses Tipps der Woche mit der Erläuterung des Einsatzes der Klassenbibliothek in der Praxis und der Funktionsweise.

Nachdem im ersten Teil dieses Tipps die Klassenbibliothek "clsExecProg" aufgebaut wurde, geht es nun in diesem Teil um den Einsatz in der Praxis und die Funktionsweise. Der Zugriff erfolgt über eine Objektvariable, die, wie im folgenden Beispiel gezeigt, in der jeweiligen VBA-Routine deklariert und initialisiert wird:

Private Sub btnStartNotepad_Click() 
Dim ep As New clsExecProg

Wichtig ist, dass Sie hier "New" angeben, damit gleich eine Instanz der Klasse initialisiert wird und sie sofort genutzt werden kann:


  With ep
    .ProgName = "notepad.exe"
    .ProgParas = "c:\autoexec.bat"
    .WorkDir = "c:\"
    .WinMode = vbMaximizedFocus
    .ExecProg
    If .EPError Then
      Beep
      MsgBox "Fehler beim Aufruf...", _
      vbOKOnly + vbInformation, "ExecProg:"
      Exit Sub
    End If

    While .IsRunning
      DoEvents 'Warten...
    Wend

    MsgBox "NotePad wurde beendet..."

  End With

End Sub

Das Beispiel öffnet die Datei "c:\autoexec.bat" zur Bearbeitung in NotePad, wartet, bis die Bearbeitung abgeschlossen, also NotePad beendet wurde und meldet dann den Abschluss der Ausführung der externen Anwendung.

Initialisierung und Aufruf erfolgen am besten innerhalb einer "With...End With"-Konstruktion, um etwas Tipparbeit zu sparen. Im Beispiel werden im ersten Schritt die im folgenden erläuterten Eigenschaften gesetzt, dann die Methode "ExecProg" für die Ausführung aufgerufen und anschließend über "EPError" geschaut, ob es ein Probleme gegeben hat. Wenn ja, erfolgt eine entsprechende Meldung und die Routine wird verlassen. Die "While...Wend"-Schleife läuft, solange die Eigenschaft "IsRunning" den Inhalt "True" hat. Dahinter verbirgt sich eine Funktion, die über API den Status der jeweiligen Anwendung abfragt. Wenn das System zurückmeldet, dass die Anwendung nicht mehr läuft, wird "IsRunning" auf "False" gesetzt und somit die Schleife verlassen. Sie können selbstverständlich auch auf die "While...Wend"-Schleife verzichten. "ExecProg" arbeitet dann wie die "Shell()"-Funktion: Die externe Anwendung wird geöffnet und die VBA-Routine sofort weiter ausgeführt.

Im Folgenden die Erläuterung der Eigenschaften, die Sie nutzen können:

ProgName: 
Definiert Laufwerk, Pfad und Dateiname der Anwendung, die gestartet werden soll, also zum Beispiel "D:\Tools\FTPTrans.exe". Sie können Laufwerk und Pfad auch weglassen, es wird dann in folgenden Verzeichnissen gesucht: Aktuelles Verzeichnis, Windows-Verzeichnis, System-Verzeichnis, Verzeichnisse, die in AUTOEXEC über PATH festgelegt wurden. Der Dateiname muss aber immer, und zwar komplett mit Erweiterung, angegeben werden! Wenn Sie nur ein Dokument mit seiner dazugehörigen Anwendung öffnen wollen, geben Sie hier nichts bzw. eine leere Zeichenkette ("") an. Achtung! Bei langen Dateinamen mit Leerzeichen muss die Zuweisung in Hochkommas eingeschlossen werden:

.ProgName = Chr$(34) + strPName + Chr$(34)

ProgParas: 
Über diese Eigenschaft legen Sie die Parameter fest, die der Anwendung übergeben werden sollen. Dies kann ein Dateiname und/oder Schalter wie beispielsweise "Z:\Daten\ebcdic.txt" /C /F=ansi" sein. Möchten Sie ein Programm "leer" starten, geben Sie hier einfach nichts bzw. eine leere Zeichenkette ("") an. Sie können hier auch nur einen Dokumentnamen wie zum Beispiel "D:\Daten\Budget.xls" angeben und "ProgName" leer lassen bzw. auf eine leere Zeichenkette setzen. "ExecProg" startet dann automatisch die Anwendung, die unter Windows für den betreffenden Dokumenten-Typ zuständig ist, in diesem Fall also beispielsweise Excel. Achtung! Bei langen Dateinamen mit Leerzeichen muss die Zuweisung in Hochkommas eingeschlossen werden:

.ProgParas = Chr$(34) + strFName + Chr$(34)

WorkDir: 
Legen Sie hier das zu verwendende Arbeitsverzeichnis fest. In der Regel handelt es sich dabei um das Verzeichnis, aus dem Sie das Programm oder ein Dokument aufrufen. Wenn Sie hier nichts angeben, wird das jeweils zuletzt aktivierte Verzeichnis auf dem betreffenden Laufwerk verwendet.

WinMode: 
Diese Eigenschaft steuert, in welchem Modus das Fenster des Programms angezeigt wird. Sie können hier die gewohnten "vb..."-Konstanten verwenden, die Sie auch bei "Shell()" benutzen:

· VbHide: Fenster ausgeblendet mit Fokus

· VbNormalFocus: Fenster in normaler Größe mit Fokus

· VbMinimizedFocus: Fenster als Symbol mit Fokus

· VbMaximizedFocus: Fenster als Vollbild mit Fokus

· VbNormalNoFocus: Fensters in normaler Größe ohne Fokus

· VbMinimizedNoFocus: Fenster als Symbol ohne Fokus

EPError: 
Diese Eigenschaft vom Datentyp "Boolean" gibt Auskunft darüber, ob die jeweils letzte Aktion erfolgreich ausgeführt werden konnte oder nicht. Bei einem Fehler steht "EPError" auf "Wahr/True". Sie sollten "EPError" zumindest nach dem Aufruf von "ExecProg" wie im Beispiel gezeigt abfragen.

Abschließend zur Funktionsweise der Klassenbibliothek: Dreh- und Angelpunkt der Lösung sind zwei API-Funktionen: "ShellExecuteEx" ist zuständig für den Start einer Anwendung. Die benötigten Informationen werden in einer Datenstruktur "SHELLEXECUTEINFO" übergeben, die von der Methode "ExecProg" anhand der gesetzten Eigenschaften gefüllt wird. "ShellExecuteEx" liefert als Ergebnis unter anderem ein Handle auf den neu gestarteten Prozess. Anhand des Handles kann man nun über "WaitForSingleObject" den Status der Anwendung erfragen. Ist die Rückmeldung größer "0", so läuft die Anwendung noch, andernfalls wurde sie inzwischen beendet. Umgesetzt für die Eigenschaft, die in einem Klassenmodul ja eigentlich eine Funktion ist, sieht das dann so aus:

Public Property Get IsRunning() As Boolean
  Dim lngStatus As Long

  lngStatus = _
  WaitForSingleObject(ep_ProcInfo.hProcess, 10)

  If lngStatus = -1 Then 'Ungültiger Handle...
    IsRunning = False
    ep_Error = True
  Else
    IsRunning = (lngStatus <> 0)
  End If

End Property

Der zweite Parameter bei "WaitForSingleObject" spezifiziert eine Zeit in Millisekunden, die auf die Beendigung der Anwendung gewartet werden soll. Wenn Sie feststellen, dass die Routine auf Ihrem System zuviel CPU-Zeit beansprucht, setzen Sie den Wert hier einfach ein wenig herunter. Nun zur Methode "ExecProg":

Public Sub ExecProg()

  If ep_ProgName <> "" Then 'Anwendung angegeben...
    ep_ShellExecInfo.cbSize = Len(ep_ShellExecInfo)
    ep_ShellExecInfo.fMask = SEE_MASK_NOCLOSEPROCESS + _
                             SEE_MASK_FLAG_NO_UI
    ep_ShellExecInfo.lpFile = ep_ProgName
    ep_ShellExecInfo.lpParameters = ep_ProgParas
    ep_ShellExecInfo.lpDirectory = ep_WorkDir
    ep_ShellExecInfo.nShow = ep_WinMode
    ep_Error = ShellExecuteEx(ep_ShellExecInfo)
    ep_ProcInfo.hProcess = ep_ShellExecInfo.hProcess
    ep_Error = (ep_ShellExecInfo.hInstApp < 32)
  Else 'Nur Datei/Dokument angegeben...
    ep_ShellExecInfo.cbSize = Len(ep_ShellExecInfo)
    ep_ShellExecInfo.fMask = SEE_MASK_NOCLOSEPROCESS + _
                             SEE_MASK_FLAG_NO_UI
    ep_ShellExecInfo.lpFile = ep_ProgParas
    ep_ShellExecInfo.lpParameters = vbNullString
    ep_ShellExecInfo.lpDirectory = ep_WorkDir
    ep_ShellExecInfo.nShow = ep_WinMode
    ep_Error = ShellExecuteEx(ep_ShellExecInfo)
    ep_ProcInfo.hProcess = ep_ShellExecInfo.hProcess
    ep_Error = (ep_ShellExecInfo.hInstApp < 32)
  End If

End Sub

Generell prüft die Routine zunächst, ob ein Programm angegeben wurde. Wenn ja, kommt der erste Teil zum Einsatz. Der zweite Teil ist zuständig für die Initialisierung, wenn nur ein Dokument zu öffnen ist. Der wesentliche Unterschied besteht darin, dass hier das Dokument sozusagen als Anwendung und das Feld für die Anwendungs-Parameter auf "Null" gesetzt wird. Die ersten vier Anweisungen "schaufeln" jeweils die gesetzten Eigenschaften in die oben erwähnte Datenstruktur, die dann als Parameter an "ShellExecuteEx" übergeben wird.

